
Generador de funciones periódicas con controlador lógico-programable (PLC)

Vidal, F., Barberà, E.*

IQS School of Engineering, Universitat Ramon Llull

Via Augusta 390, 08017 - Barcelona, Spain

Generador de funcions periòdiques amb controlador lògic-programable (PLC)

Generator-periodic functions with programmable logic controller (PLC)

Recibido: 6 de noviembre de 2014; aceptado: 13 de marzo de 2015

RESUMEN

El análisis frecuencial de sistemas de control se basa en el conjunto de respuestas del proceso a señales senoidales que, en la mayoría de casos de los procesos habituales en la industria química, han de ser de muy baja frecuencia. Los generadores de señal comunes en los laboratorios de electrónica no llegan a las bajas frecuencias necesarias, por tanto se ha considerado oportuno programar un controlador lógico-programable (PLC) para su generación y se ha ampliado a señales rectangulares, triangulares y en diente de sierra.

La función programada se ha aplicado a la obtención de la respuesta frecuencial de un sistema de control de nivel.

Palabras clave: Controladores lógico-programables, generación de señales, respuesta frecuencial

SUMMARY

Frequency-response analysis of control systems is based on process response to sinusoidal inputs that, in chemical processes, should be of very low frequencies. Common signal generators do not provide very low frequencies, therefore, a programmable logic controller (PLC) has been

programmed in order to obtain the sinusoidal signal. In addition, rectangular, triangular and sawtooth waves have been considered.

This function has been applied to frequency-response analysis of a level control system.

Keywords: Programmable logic controllers, signal generation, frequencial response

RESUM

L'anàlisi freqüencial de sistemes es basa en el conjunt de respostes del procés a senyals sinusoidals que, normalment en els processos habituals a l'indústria química, han de ser de molt baixa freqüència. Els generadors de senyal habituals en els laboratoris d'electrònica no arriben a les baixes freqüències necessàries, per tant s'ha considerat adient programar un controlador lògic programable (PLC) per la seva generació i s'ha ampliat a senyals rectangulars, triangulars i en dent de serra.

La funció programada s'ha aplicat a la obtenció de la resposta freqüencial d'un sistema de control de nivell.

Paraules clau: Controladors lògic-programables, generació de senyals, resposta freqüencial

INTRODUCCIÓN

El análisis frecuencial de sistemas de control requiere la obtención de la respuesta del proceso a señales senoidales que, en la mayoría de casos de procesos químicos, han de ser de muy baja frecuencia. De esta manera se puede obtener la respuesta frecuencial que, además de proporcionar información sobre las características del comportamiento del proceso, permite diseñar el control del mismo.

Los generadores de señales habituales en los laboratorios de electrónica no alcanzan las bajas frecuencias necesarias, por tanto ha parecido oportuno utilizar un equipo de uso común tal como un controlador lógico-programable, que además tiene las características de una gran fiabilidad y robustez en su funcionamiento.

Se pretende programar una función que permita la obtención de diversas señales: senoidal, cuadrada, triangular y en diente de sierra con la amplitud y valor medio deseados.

GENERACIÓN DE LAS SEÑALES

Primero se evalúan todas las señales centradas en cero y comprendidas en el intervalo entre -1 y 1 y posteriormente se escalarán a los valores deseados.

Para la evaluación de las diversas funciones se calcula un tiempo que se reinicia en 0 cada vez que transcurre un periodo (TP).

$$TP = \frac{2\pi}{\omega}$$

donde ω es la frecuencia de la señal en radianes por segundo.

Señal senoidal (y_1)

Se ha escogido la función senoidal para generar la señal básica, utilizando la ecuación siguiente:

$$y_1 = \text{sen}(\omega t)$$

donde ω es la frecuencia en radianes por segundo.

Onda cuadrada (y_2)

La onda cuadrada (Figura 1) se obtiene a partir de la evaluación del signo de la función senoidal.

$$y_2 = \text{signo}(y_1)$$




Figura 1.- Onda cuadrada

Onda triangular (y_3)

La onda triangular simétrica sigue la ecuación de una recta cuya pendiente es alternativamente positiva y negativa, aunque es la misma en valor absoluto (Figura 2).




Figura 2.- Onda triangular

El valor absoluto de la pendiente calculada a partir de que la señal recorre un intervalo de valor 2 (entre -1 y 1) en medio periodo, es:

$$a_3 = \frac{4}{TP}$$

En la subida ($0 < t < TP/2$) la ecuación es:

$$y_3 = \frac{4}{TP}t - 1 \quad \text{si } y_2 > 0$$

En el descenso ($TP/2 < t < TP$):

$$y_3 = -\frac{4}{TP}t + 3 \quad \text{si } y_2 < 0$$

Señal en diente de sierra (y_4)

Se ha escogido únicamente una señal de pendiente positiva, por lo que la señal va desde -1 hasta 1 durante un periodo (Figura 3), por tanto el valor de la pendiente es:

$$a_4 = \frac{2}{TP}$$




Figura 3.- Onda en diente de sierra

El cálculo de la señal se realiza con la siguiente expresión:

$$y_4 = \frac{2}{TP}t - 1$$

Escalado

Una vez evaluadas las señales entre -1 y 1, se procede a su escalado utilizando la siguiente expresión.

$$\text{salida} = By_i + A$$

donde B es la amplitud de la señal y A el valor medio.

Al escalar se debe tener en cuenta el margen de salida real del convertidor analógico-digital del PLC para evitar la saturación del mismo.

Programa

Se ha desarrollado un bloque de función (FB4) desarrollado para un PLC Siemens S-300, concretamente la CPU compacta 314C-2DP con entradas y salidas analógicas de 12 bits utilizando el programa Step 7® Professional 2006 (1).

La tabla de variables de la función (Tabla 1), coincidente con el bloque de datos DB asociado, es:

Tabla 1.- Variables del DB asociado a la función FB4

IN	W	real	frecuencia radianes/segundo
IN	Ts	real	periodo de cálculo
IN	A	real	valor medio
IN	B	real	amplitud
OUT	Y1	real	senoidal
OUT	Y2	real	cuadrada
OUT	Y3	real	triangular
OUT	Y4	real	sierra
STAT	per_time	real	tiempo dentro del periodo

La variable per_time indica el tiempo comprendido en un periodo de la señal, esta variable se incrementa cada vez que se ejecuta el bloque de organización de interrupción cíclica, OB35, con un valor igual al periodo de ejecución de dicho bloque. Cuando per_time supera el valor del periodo de la señal, se resta este valor para conseguir que el valor de per_time esté siempre situado entre 0 y el valor del periodo de la señal.

El programa se ha escrito de forma que sea muy simple su modificación, recortando las porciones innecesarias para cumplir con una finalidad particular.

```

FB4 "generador_funciones"
//cálculo del tiempo
L #per_time //tiempo dentro del periodo
L #Ts //periodo de cálculo
+R
T #per_time
L 2.000000e+000
L 3.141590e+000
*R
L #W //frecuencia (rad/s)
/R //periodo de la señal, TP
L #per_time
>R
SPB eti1
TAK //intercambio de acumuladores
-R //si tiempo > periodo se resta
T #per_time
eti1: NOP 0
//senoidal
L #per_time
L #W
*R //wt
SIN
T #Y1
//cuadrada
L #Y1 //señal senoidal
L 0.000000e+000
>=R //evaluación del signo
SPB eti2
L -1.000000e+000 //signo negativo
SPA eti3
eti2: L 1.000000e+000 //signo positivo
eti3: T #Y2
//triangular
L #Y1 //señal senoidal
L 0.000000e+000
>=R //evaluación del signo
SPB eti4
L 2.000000e+000
L 3.141590e+000
*R
L #W
/R //periodo
L -4.000000e+000
TAK //intercambio de acumuladores
/R //pendiente bajada, -a3
L #per_time
*R
L 3.000000e+000
+R //salida, y3, bajada
SPA eti5
eti4: L 2.000000e+000
L 3.141590e+000

```

```

*R
L #W
/R //periodo
L 4.000000e+000
TAK //intercambio de acumuladores
/R //pendiente subida, a3
L #per_time
*R
L -1.000000e+000
+R //salida, y3, subida
eti5: T #Y3
//diente de sierra
L 2.000000e+000
L 3.141590e+000
*R
L #W
/R //periodo
L 2.000000e+000
TAK //intercambio de acumuladores
/R //pendiente subida, a4
L #per_time
*R
L -1.000000e+000
+R
T #Y4
//escalado senoidal
L #B //amplitud, B
L #Y1
*R
L #A //desplazamiento, A
+R
T #Y1
//escalado cuadrada
L #B //amplitud, B
L #Y2
*R
L #A //desplazamiento, A
+R
T #Y2
//escalado triangular
L #B //amplitud, B
L #Y3
*R
L #A //desplazamiento, A
+R
T #Y3
//escalado diente de sierra
L #B //amplitud, B
L #Y4
*R
L #A //desplazamiento, A
+R
T #Y4

```

Tal como se puede apreciar, el periodo de la señal se calcula en cada segmento porque se ha optado por minimizar las variables internas utilizadas, en detrimento de la longitud del programa y del tiempo de ejecución. En el caso en que se quisiera optimizar el tiempo y la longitud de programa bastaría evaluar el periodo una única vez y almacenarlo en una variable. También se acortaría el programa creando una función para el escalado final de las señales. Para comprobar el buen funcionamiento del bloque desarrollado se ha generado un programa principal, OB1, que se limita a llamar la función "UNSCALE", que transforma la

señal real en un entero preparado para enviarlo a una salida analógica, en este caso la salida de periferia PAW752 que, en el entrenador utilizado, está conectada a un voltímetro. En este caso se escala entre +10V y -10V, intervalo que coincide con el margen de salida configurado en el convertidor analógico-digital.

El valor de la señal generada se almacena en la marca MD18.

OB1

```
CALL "UNSCALE" //salida
IN :=MD18 //valor de la señal generada
HI_LIM :=1.000000e+001//límite superior de la señal
LO_LIM :=-1.000000e+001//límite inferior de la señal
BIPOLAR:=TRUE
RET_VAL:=MW22
OUT :=PAW752 //salida analógica
```

Para conseguir una correcta temporización, la llamada al bloque desarrollado, FB4, se realiza dentro del bloque de organización de interrupción cíclica, OB35, que en este caso se ha configurado con un ciclo de 100ms. El valor del ciclo aparece en la variable #OB35_EXC_FREQ, este valor se pasa a segundos y se almacena en la marca MD10.

OB35 "Cyclic interrupt"

```
L #OB35_EXC_FREQ //periodo de ejecución en ms
ITD
DTR
L 1.000000e+003
/R
T MD 10
CALL FB 4, DB4
W :=1.000000e-001 //frecuencia
Ts:=MD10 //periodo de ejecución
A := 0.000000e+000 //desplazamiento
B := 1.000000e+000 //amplitud
Y1:= //onda senoidal
Y2:= //onda cuadrada
Y3:= //onda triangular
Y4:=MD18 //onda diente de sierra
```

En el bloque anterior, se pretende generar una onda en diente de sierra (Y4) entre +1.0 y -1.0, con una frecuencia de 0.1 rad/s y un periodo de muestreo de 0.1 s (coincidente con el ciclo de ejecución del OB35). En el caso de querer la generación de diferentes tipos de onda de la misma frecuencia y con el mismo escalado, basta asociar las salidas oportunas a diferentes marcas.

Si se desea obtener al mismo tiempo y con el mismo equipo señales de diferente frecuencia o diferente escalado basta con llamar las veces necesarias al bloque de función con una DB diferente que contenga los valores deseados para cada caso.

APLICACIÓN A LA OBTENCIÓN DE LA RESPUESTA FRECUENCIAL DE UN PROCESO

La respuesta frecuencial consiste en el conjunto de respuestas en estado estacionario a entradas senoidales de frecuencia variable. El proceso consiste en un depósito

con un caudal de líquido de entrada regulado por una válvula y un sensor de nivel (Figura 4).




Figura 4.- Diagrama del proceso

MATERIALES

- convertidor I/P: SAUTER XEP10 F002: convierte la señal de salida del controlador (4 a 20 mA) en una señal de presión de 0.2 a 10 bar.
- actuador: SAUTER V6R (válvula) + SAUTER AV42P (motor neumático).
- medidor de nivel: MOBREY MSP422: la señal de salida es de intensidad 4 a 20 mA y se debe parametrizar introduciendo el valor de nivel o distancia que corresponde a 4 mA y a 20 mA, en este caso 0 a 43 cm.
- medidor de caudal: SIGNET 2536 (sensor) + SIGNET 8512 (convertidor frecuencia / intensidad): la señal que proporciona es una intensidad entre 4 y 20 mA, la presentación en la pantalla dependerá de los valores asignados.
- convertidor I/V: BRODERSEN PXU-20.230 (3 unidades): se utiliza para convertir de forma eficiente la señal de intensidad (4 a 20 mA) a una señal de voltaje (0 a 10 V) directamente conectable a registrador o a interfase de ordenador.
- interfase KUSB 3100 de Keithley, acompañada del software necesario (QuickDataAcq V 1.0.6 desarrollado por F. Consiglio para Data Translation Inc.). Este programa permite la captación de datos y su almacenamiento en el formato Excel® (Microsoft), lo que facilita el posterior tratamiento.

Procedimiento

La función desarrollada se ha aplicado a la obtención de la respuesta frecuencial de un proceso formado por una válvula y un depósito por el procedimiento de barrido (2), es decir: se introduce una señal senoidal de amplitud y frecuencia determinadas como variable de entrada y se captura la respuesta del proceso, repitiendo el procedimiento para todas las frecuencias deseadas. La ganancia del proceso para una frecuencia determinada se obtiene dividiendo la amplitud de la respuesta por la amplitud de la entrada y el desfase consiste en la diferencia de fase entre la respuesta y la entrada. Explorando el margen deseado de frecuencias podrá representarse la respuesta frecuencial del proceso. La existencia de no linealidades en el proceso podrá dificultar la interpretación de los resultados y la obtención de un modelo a partir de la respuesta frecuencial. Aunque las salidas analógicas del PLC Siemens 314C-2DP permiten su configuración como salidas de potencial o de intensidad, se ha preferido respetar la configuración habitual de potencial y transformar éste en intensidad mediante una resistencia en serie de 499 Ω. Al utilizar esta transformación tan simple se debe tener en cuenta que una salida de 10 V no podrá dar una intensidad de 20 mA, dado que se debe considerar también la resistencia de entrada del equipo receptor. La señal senoidal generada por el PLC y transformada en intensidad pasa a un convertidor intensidad / presión que hace mover la válvula modificando el caudal que llega al depósito modificando el nivel de líquido en el mismo, éste es medido por un sensor de nivel que proporciona una señal de intensidad que se convierte a diferencia de potencial. También se dispone de un sensor de caudal cuya señal,

convenientemente convertida a diferencia de potencial, se recoge junto con el nivel y la salida del PLC en la interfase de adquisición de datos, KUSB 3100 de Keithley. Para generar la onda senoidal, a utilizar en la obtención de la respuesta frecuencial del proceso, se deben modificar los valores de amplitud y desplazamiento (parámetros B y A del bloque de función) siguiendo los criterios mencionados anteriormente.




Figura 5.- Diagrama de conexiones para realizar la respuesta a las entradas senoidales.

Con el fin de disponer de un número suficiente de puntos para dibujar el diagrama de Bode y dada la naturaleza logarítmica del eje de abscisas se ha trabajado con la siguiente tabla de frecuencias:

Tabla 2.- Tabla de frecuencias

frecuencia, ω
1.0000000
1.2589254
2.5118864
3.9810717
5.0118723
6.3095734
7.9432823

Cada valor de frecuencia irá afectado por la potencia de 10 correspondiente a la década en que se esté trabajando, siendo la frecuencia mínima de 0.001000 y la máxima de 10.00 rad/s.

Resultados

Como ejemplo se presentan los datos adquiridos para tres frecuencias, a 0.001, 0.1 y 1 rad/s (Figuras 6 a 8).




Figura 6.- Datos adquiridos (V) en función del tiempo (s) a $\omega = 0.001$ rad/s




Figura 7.- Datos adquiridos (V) en función del tiempo (s) a $\omega = 0.1$ rad/s




Figura 8.- Datos adquiridos (V) en función del tiempo (s) a $\omega = 1$ rad/s

Se puede apreciar que al aumentar la frecuencia, la amplitud de las señales de caudal y de nivel disminuyen, a frecuencia igual a 1 rad/s el nivel aparece constante (Figura 8), dado el comportamiento como filtro pasabajos del proceso. El gráfico del caudal que alimenta al depósito en función de la posición de la válvula presenta, a baja frecuencia un aspecto claramente deformado debido a la no linealidad de la válvula. Esta no linealidad se aprecia muy claramente al representar (Figura 9) el caudal en función de la señal de entrada a la válvula (salida del PLC), dado que no se obtiene una elipse perfecta, que sería el resultado si la válvula tuviera un comportamiento lineal.




Figura 9.- Caudal (V) en función de la señal de entrada (V) a la válvula

Para las señales obtenidas para cada frecuencia se ha ajustado una onda senoidal con el fin facilitar la evaluación de la amplitud y del desfase. A partir de estos valores de

amplitud y desfase se han obtenido los correspondientes valores de ganancia y argumento de la respuesta frecuencial con el fin de dibujar su diagrama de Bode.

Se presenta el diagrama de Bode del proceso constituido por la válvula y el medidor de caudal (Figura 10), cuya entrada es la señal de salida del PLC y la respuesta es el caudal.




Figura 10.- Diagrama de Bode del proceso válvula-medidor de caudal. Módulo en dB y argumento en grados frente a la frecuencia en rad/s.

Igualmente se han obtenido los diagramas de Bode correspondientes a los procesos nivel/caudal y válvula/nivel, estos resultados no se presentan por no aportar una información significativa.

CONCLUSIONES

Se ha obtenido un bloque de función que permite que un PLC actúe como un generador de señales de muy baja frecuencia, lo que posibilita llevar a cabo el análisis de la respuesta de procesos reales frente a las señales generadas. En particular, se puede obtener la respuesta frecuencial utilizando señales senoidales, pudiendo llegar a dibujar el diagrama de Bode o de Nyquist del proceso.

Se ha aplicado el bloque desarrollado a la obtención de la respuesta frecuencial de un proceso compuesto por una válvula y un depósito con medición de caudal y de nivel.

REFERENCIAS

1. Molins, J.J.; Barberà, E.: "Autómatas programables: Step7® y UnityPro®", IQS, 2012
2. Moles, S.; Barberà, E.: "Ampliación y mejora de una práctica de control de nivel", 12th Mediterranean Congress of Chemical Engineering, Barcelona, 15-18 noviembre 2011